

## Example of Crossed Random Effects Models: Eye Movements nested within Subjects and within Items

Source: Mills, M., Hollingworth, A., Van der Stigchel, S., Hoffman, L., & Dodd, M. D. (2011). Examining the influence of task set on eye movements and fixations. *Journal of Vision*, 11(8), 1-15.

Fixation duration data for a scene viewing task was collected for 67 items (scenes) from 53 subjects (total observations used = 60,709). No scene predictors were of interest, but subjects varied by the visual task they were assigned to do: Free View, Memorize, Rate Pleasantness, or Search.

```
* Fixation data - creating and labeling new variables;
DATA Work.FixTrim; SET work.Fixate;
  FixTime1s=FixTime - 1; FixTime2s=FixTime - 2; FixTime5s=FixTime - 5;
  FixTimeRd=ROUND(FixTime, .50);
  LABEL FixTime1s="FixTime1s: Start Time of Fixation in Seconds (0=1s)"
        FixTime2s="FixTime2s: Start Time of Fixation in Seconds (0=2s)"
        FixTime5s="FixTime5s: Start Time of Fixation in Seconds (0=5s)"
        FixTimeRd="FixTimeRd: Start Time of Fixation Rounded to 200ms"
        FixDur= "FixDur: Fixation Duration (90-2000ms) No Blink"
        cond= "cond: Viewing Condition (Free, Memory, Pleasant, Search)";
RUN;
* Removing first fixation duration;
DATA Work.FixTrim; SET Work.FixTrim; WHERE FixIndex>1; RUN;
```

---

### Model 1: Estimate baseline empty means model with only residual variance

(default REPEATED statement if not included is TYPE=VC)  $\text{FixDur}_{tis} = \gamma_{000} + \epsilon_{tis}$

```
TITLE1 "Fixation Duration Empty model - no random effects";
PROC MIXED DATA=Work.FixTrim NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS PersonID ItemID cond;
  MODEL FixDur = / SOLUTION DDFM=SATTERTHWAITE;
RUN;
```

Dimensions	
Covariance Parameters	1
Columns in X	1
Columns in Z	0
Subjects	1
Max Obs Per Subject	69377

Number of Observations	
Number of Observations Read	69377
Number of Observations Used	60709
Number of Observations Not Used	8668

Covariance Parameter Estimates				
Cov Parm	Estimate	Standard Error	Z	Pr > Z
Residual	18406	105.65	174.23	<.0001

All the variance in FixDur in one pile of e (TYPE=VC)

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
768,474	2	768478	768478	768484	768496	768498

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	256.19	0.5506	61E3	465.26	<.0001

## Model 2: Is there significant systematic variation across subjects? $\text{FixDur}_{tis} = \gamma_{000} + U_{00s} + e_{tis}$

```
TITLE1 "Fixation Duration Empty model - random persons";
PROC MIXED DATA=Work.FixTrim NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
CLASS PersonID ItemID cond;
MODEL FixDur = / SOLUTION DDFM=SATTERTHWAITE;
RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID; RUN;
```

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	PersonID	1098.26	214.79	5.11	<.0001	Subject Intercept Variance
Residual		17384	99.8248	174.15	<.0001	Leftover fixation-to-fixation variance

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
765,234	3	765240	765240	765242	765245	765248

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	259.07	4.5844	53.8	56.51	<.0001

### Is model 2 better than model 1? How do we know?

Variance:  $1,098 / 18,482 = 5.9\%$  is between subjects,  $17,384 / 18,482 = 94.1\%$  is within subjects

## Model 3: Is there significant systematic variation across items? $\text{FixDur}_{tis} = \gamma_{000} + U_{00s} + U_{0i0} + e_{tis}$

```
TITLE1 "Fixation Duration Empty model - random persons and items";
PROC MIXED DATA=Work.FixTrim NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
CLASS PersonID ItemID cond;
MODEL FixDur = / SOLUTION DDFM=SATTERTHWAITE;
RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;
RANDOM INTERCEPT / TYPE=UN SUBJECT=ItemID; RUN;
```

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	PersonID	1115.09	219.90	5.07	<.0001	Subject Intercept Variance
UN(1,1)	ItemID	53.0516	12.4418	4.26	<.0001	Item Intercept Variance
Residual		17328	99.5557	174.06	<.0001	Leftover fixation-to-fixation variance

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
765,126	4	765134	765134	765126	765126	765130

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	258.93	4.7041	56.8	55.04	<.0001

### Is model 3 better than model 2? How do we know?

Variance:  $1,115 / 18,496 = 6.0\%$  is between subjects

$53 / 18,496 = 0.3\%$  is between items

$17,328 / 18,496 = 93.7\%$  is within subjects and items (subject x item interaction)

So what does this imply for predicting SUBJECT group differences due to task?


And what are we trying to model here, again? Let's plot the saturated means from rounded time...

```
TITLE1 "Fixation Duration Saturated means model - random persons and items";
PROC MIXED DATA=Work.FixTrim NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS PersonID ItemID cond FixTimeRd;
  MODEL FixDur = FixTimeRd cond FixTimeRd*cond / SOLUTION DDFM=SATTERTHWAITE;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=ItemID;
  LSMEANS FixTimeRd*cond;
  ODS OUTPUT LSMEANS=FixDur_means;
RUN;
* Word file graphs get saved to, NO page breaks between plots;
ODS RTF FILE="&filesave.\Fixation Duration Means by cond.rtf" STARTPAGE=NO;
TITLE1 JUSTIFY=CENTER HEIGHT=1.5 "Fixation Duration Means by cond";
PROC GPLOT DATA=FixDur_means;
  * Settings for Y-axis;
  AXIS1 LENGTH=3.0in LABEL=(ANGLE=90 HEIGHT=1.0 "Fixation Duration") ORDER=(150 TO 350 BY 20);
  * Settings for X-axis;
  AXIS2 LENGTH=5.0in LABEL=(HEIGHT=1.0 "Viewing Time (ms)") ORDER=(0 TO 6 BY .5);
  * Settings for lines;
  SYMBOL1 VALUE=DOT COLOR=RED HEIGHT=1.3 LINE=1 WIDTH=2;
  SYMBOL2 VALUE=DOT COLOR=GREEN HEIGHT=1.3 LINE=1 WIDTH=2;
  SYMBOL3 VALUE=DOT COLOR=BLUE HEIGHT=1.3 LINE=1 WIDTH=2;
  SYMBOL4 VALUE=DOT COLOR=PURPLE HEIGHT=1.3 LINE=1 WIDTH=2;
  * Settings for legend;
  LEGEND1 NOFRAME POSITION=(CENTER TOP) ACROSS=2 LABEL=NONE VALUE=(COLOR=Black HEIGHT=1.3
    "Freeview" "Memory" "Pleasantness" "Search");
  * Plot Y*X=separate person, start axes at 0 unless otherwise specified;
  PLOT Estimate*FixTimeRd=cond / LEGEND=LEGEND1 VAXIS=AXIS1 HAXIS=AXIS2;
RUN; QUIT; TITLE1; ODS RTF CLOSE;
```

Fixation Duration Means by cond

●—● Freeview      ●—● Memory  
●—● Pleasantness      ●—● Search

Looks quadratic-y to me,  
with a lot of noise in the  
first 500 ms or so...



**Model 4a: Fixed Quadratic, Random Intercept Effects of Viewing Time**

$$\text{FixDur}_{tis} = \gamma_{000} + \gamma_{100}(\text{Time}_{tis}-1) + \gamma_{200}(\text{Time}_{tis}-1)^2 + U_{00s} + U_{0i0} + \epsilon_{tis}$$

I made "time" a macro variable so I could easily use a different version to get simple effects at particular times of interest.

```
* Time metric; %LET time=FixTime1s; %LET center=1sec;
```

```
TITLE1 "Fixation Duration Fixed Quadratic, Random Intercept &time. Model";
PROC MIXED DATA=Work.FixTrim NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS PersonID ItemID cond;
  MODEL FixDur = &time. &time.*&time. / SOLUTION DDFM=SATTERTHWAITE;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=ItemID;
  PARS (1115) (53) (17328); * Start values for variances; RUN;
```

## Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	PersonID	1111.25	217.43	5.11	<.0001
UN(1,1)	ItemID	48.4782	11.7756	4.12	<.0001
Residual		17091	98.1926	174.05	<.0001

## Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
764,285	6	764297	764297	764285	764285	764291

## Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	238.32	4.7454	60.1	50.22	<.0001
FixTime1s	14.5782	0.9450	6E4	15.43	<.0001
FixTime1s*FixTime1s	-1.3293	0.2112	58E3	-6.29	<.0001

Calculate Pseudo-R<sup>2</sup> for residual variance →  $(17,328 - 17,091) / 17,328 = 1.4\%$  ☹

**Model 4b: Fixed Quadratic, Random Linear Effects of Viewing Time across Subjects**

$$\text{FixDur}_{tis} = \gamma_{000} + \gamma_{100}(\text{Time}_{tis}-1) + \gamma_{200}(\text{Time}_{tis}-1)^2 + U_{00s} + U_{10s}(\text{Time}_{tis}-1) + U_{0i0} + \epsilon_{tis}$$

```
TITLE1 "Fixation Duration Fixed Quadratic, Random Linear &time. Model";
PROC MIXED DATA=Work.FixTrim NOCLPRINT NOITPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS PersonID ItemID cond;
  MODEL FixDur = &time. &time.*&time. / SOLUTION DDFM=SATTERTHWAITE;
  RANDOM INTERCEPT &time. / TYPE=UN SUBJECT=PersonID;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=ItemID;
RUN;
```

## Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr Z	
UN(1,1)	PersonID	935.14	189.68	4.93	<.0001	Subject Intercept Variance
UN(2,1)	PersonID	10.0839	26.9714	0.37	0.7085	Subject Intercept-Linear Covariance
UN(2,2)	PersonID	33.9319	7.6434	4.44	<.0001	Subject Linear Time Variance
UN(1,1)	ItemID	49.0094	11.8590	4.13	<.0001	Item Intercept Variance
Residual		16994	97.6812	173.98	<.0001	Everything else

## Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
764,041	8	764057	764057	764041	764041	764049

Better model than 4a?

## Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	238.36	4.3836	57.5	54.38	<.0001
FixTime1s	14.2695	1.2402	226	11.51	<.0001
FixTime1s*FixTime1s	-1.2125	0.2113	59E3	-5.74	<.0001

**Model 4c: Random Quadratic Effects of Viewing Time across Subjects**

$$\text{FixDur}_{tis} = \gamma_{000} + \gamma_{100}(\text{Time}_{tis}-1) + \gamma_{200}(\text{Time}_{tis}-1)^2 + U_{00s} + U_{10s}(\text{Time}_{tis}-1) + U_{20s}(\text{Time}_{tis}-1)^2 + U_{0i0} + e_{tis}$$

```
TITLE1 "Fixation Duration Random Quadratic &time. Model";
PROC MIXED DATA=Work.FixTrim NOCLPRINT NOITPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS PersonID ItemID cond;
  MODEL FixDur = &time. &time.*&time. / SOLUTION DDFM=SATTERTHWAITE;
  RANDOM INTERCEPT &time. &time.*&time. / TYPE=UN SUBJECT=PersonID;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=ItemID;
RUN;
```

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z	Your turn!
UN(1,1)	PersonID	866.15	177.58	4.88	<.0001	
UN(2,1)	PersonID	91.7966	56.1242	1.64	0.1019	
UN(2,2)	PersonID	126.56	35.1997	3.60	0.0002	
UN(3,1)	PersonID	-17.4447	13.2431	-1.32	0.1878	
UN(3,2)	PersonID	-26.8495	7.8659	-3.41	0.0006	
UN(3,3)	PersonID	7.4437	1.9481	3.82	<.0001	
UN(1,1)	ItemID	49.5432	11.9347	4.15	<.0001	
Residual		16949	97.4671	173.90	<.0001	

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
763,947	11	763969	763969	763947	763947	763958

Better model than 4b?

Solution for Fixed Effects						
Effect	Estimate	Standard Error	DF	t Value	Pr >  t	
Intercept	238.04	4.2338	57.1	56.22	<.0001	
FixTime1s	14.9089	1.8175	49.1	8.20	<.0001	
FixTime1s*FixTime1s	-1.3497	0.4313	51	-3.13	0.0029	

**Model 5a: Viewing Condition\*Quadratic Time Fixed Effects with Random Quadratic Time...**

$$\text{FixDur}_{tis} = \gamma_{000} + \gamma_{100}(\text{Time}_{tis}-1) + \gamma_{200}(\text{Time}_{tis}-1)^2 + \gamma_{001}(\text{Cond1}_s) + \gamma_{002}(\text{Cond2}_s) + \gamma_{003}(\text{Cond3}_s) + \gamma_{101}(\text{Cond1}_s)(\text{Time}_{tis}-1) + \gamma_{102}(\text{Cond2}_s)(\text{Time}_{tis}-1) + \gamma_{103}(\text{Cond3}_s)(\text{Time}_{tis}-1) + \gamma_{201}(\text{Cond1}_s)(\text{Time}_{tis}-1)^2 + \gamma_{202}(\text{Cond2}_s)(\text{Time}_{tis}-1)^2 + \gamma_{203}(\text{Cond3}_s)(\text{Time}_{tis}-1)^2 + U_{00s} + U_{10s}(\text{Time}_{tis}-1) + U_{20s}(\text{Time}_{tis}-1)^2 + U_{0i0} + e_{tis}$$

```
TITLE1 "Fixation Duration Random Quadratic at &center. by Condition Model";
PROC MIXED DATA=Work.FixTrim NOCLPRINT NOITPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS PersonID ItemID cond;
  MODEL FixDur = &time. &time.*&time. cond cond*&time. cond*&time.*&time.
    / SOLUTION DDFM=SATTERTHWAITE OUTPM=Conditional;
  RANDOM INTERCEPT &time. &time.*&time. / TYPE=UN SUBJECT=PersonID;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=ItemID;
  PARS (866) (92) (126) (-17) (-26) (7) (49) (16949); * Start values for variances;
RUN;
```

Covariance Parameter Values		
At Last Iteration		
Cov Parm	Subject	Estimate
UN(1,1)	PersonID	528.65
UN(2,1)	PersonID	59.5509
UN(2,2)	PersonID	93.2085
UN(3,1)	PersonID	-8.9670
UN(3,2)	PersonID	-20.4961
UN(3,3)	PersonID	6.1880
UN(1,1)	ItemID	26.5182
Residual		17030

There's no error message in the log, but it couldn't finish, even after adding some start values via the PARMS statement.

So it's broken, and we (really) can't keep it. Remember, random effects break models, not fixed effects...

So we back the model down to random linear and try again.

**Model 5b: Condition\*Quadratic Time Fixed Effects with Random Linear Time...** 1 = Freeview, 2 = Memory, 3 = Pleasantness, 4 = Search

$$\text{FixDur}_{tis} = \gamma_{000} + \gamma_{100}(\text{Time}_{tis}-1) + \gamma_{200}(\text{Time}_{tis}-1)^2 + \gamma_{001}(\text{Cond1}_s) + \gamma_{002}(\text{Cond2}_s) + \gamma_{003}(\text{Cond3}_s) + \gamma_{101}(\text{Cond1}_s)(\text{Time}_{tis}-1) + \gamma_{102}(\text{Cond2}_s)(\text{Time}_{tis}-1) + \gamma_{103}(\text{Cond3}_s)(\text{Time}_{tis}-1) + \gamma_{201}(\text{Cond1}_s)(\text{Time}_{tis}-1) + \gamma_{202}(\text{Cond2}_s)(\text{Time}_{tis}-1) + \gamma_{203}(\text{Cond3}_s)(\text{Time}_{tis}-1) + U_{00s} + U_{10s}(\text{Time}_{tis}-1) + U_{0i0} + e_{tis}$$

```

TITLE1 "Fixation Duration Fixed Quadratic, Random Linear at &center. by Condition Model";
PROC MIXED DATA=Work.FixTrim NOCLPRINT NOITPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS PersonID ItemID cond;
  MODEL FixDur =      &time. &time.*&time. cond cond*&time. cond*&time.*&time.
                    / SOLUTION DDFM=SATTERTHWAITE OUTPM=Conditional;
  RANDOM INTERCEPT &time. / TYPE=UN SUBJECT=PersonID;
  RANDOM INTERCEPT      / TYPE=UN SUBJECT=ItemID;
  ODS OUTPUT Tests3=FixDur_&time._Ftests ESTIMATES= FixDur_&time._Estimates; * Save all my results!;
* Group differences in intercept at time;
ESTIMATE "Intercept at &center.: Free-View"          Int 1 &time. 0 &time.*&time. 0 cond 1 0 0 0 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Memory"              Int 1 &time. 0 &time.*&time. 0 cond 0 1 0 0 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Pleasantness"        Int 1 &time. 0 &time.*&time. 0 cond 0 0 1 0 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Search"              Int 1 &time. 0 &time.*&time. 0 cond 0 0 0 1 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Free-View vs Memory" cond -1 1 0 0 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Free-View vs Pleasantness" cond -1 0 1 0 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Free-View vs Search"  cond -1 0 0 1 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Memory vs Pleasantness" cond 0 -1 1 0 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Memory vs Search"     cond 0 -1 0 1 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Intercept at &center.: Pleasantness vs Search" cond 0 0 -1 1 &time.*cond 0 0 0 0 &time.*&time.*cond 0 0 0 0;
* Group differences in linear slope at time 0;
ESTIMATE "Linear at &center.: Free-View"              &time. 1 &time.*&time. 0 &time.*cond 1 0 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Memory"                  &time. 1 &time.*&time. 0 &time.*cond 0 1 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Pleasantness"            &time. 1 &time.*&time. 0 &time.*cond 0 0 1 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Search"                  &time. 1 &time.*&time. 0 &time.*cond 0 0 0 1 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Free-View vs Memory"    &time.*cond -1 1 0 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Free-View vs Pleasantness" &time.*cond -1 0 1 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Free-View vs Search"    &time.*cond -1 0 0 1 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Memory vs Pleasantness" &time.*cond 0 -1 1 0 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Memory vs Search"       &time.*cond 0 -1 0 1 &time.*&time.*cond 0 0 0 0;
ESTIMATE "Linear at &center.: Pleasantness vs Search" &time.*cond 0 0 -1 1 &time.*&time.*cond 0 0 0 0;
* Group differences in quadratic slope for all seconds;
ESTIMATE "Quadratic for all Time: Free-View"          &time.*&time. 1 &time.*&time.*cond 1 0 0 0;
ESTIMATE "Quadratic for all Time: Memory"            &time.*&time. 1 &time.*&time.*cond 0 1 0 0;
ESTIMATE "Quadratic for all Time: Pleasantness"      &time.*&time. 1 &time.*&time.*cond 0 0 1 0;
ESTIMATE "Quadratic for all Time: Search"            &time.*&time. 1 &time.*&time.*cond 0 0 0 1;
ESTIMATE "Quadratic for all Time: Free-View vs Memory" &time.*&time.*cond -1 1 0 0;
ESTIMATE "Quadratic for all Time: Free-View vs Pleasantness" &time.*&time.*cond -1 0 1 0;
ESTIMATE "Quadratic for all Time: Free-View vs Search" &time.*&time.*cond -1 0 0 1;
ESTIMATE "Quadratic for all Time: Memory vs Pleasantness" &time.*&time.*cond 0 -1 1 0;
ESTIMATE "Quadratic for all Time: Memory vs Search"   &time.*&time.*cond 0 -1 0 1;
ESTIMATE "Quadratic for all Time: Pleasantness vs Search" &time.*&time.*cond 0 0 -1 1;
* Export saved results to excel;
PROC EXPORT DATA=FixDur_&time._Ftests   OUTFILE= "&filesave.\Fixation Duration &filename..xls" DBMS=EXCEL REPLACE; SHEET= "Ftests_&center."; RUN;
PROC EXPORT DATA=FixDur_&time._Estimates OUTFILE= "&filesave.\Fixation Duration &filename..xls" DBMS=EXCEL REPLACE; SHEET= "Estimates_&center.";
RUN;
* Get total R2;
PROC CORR DATA=Conditional; VAR pred FixDur; RUN;

```

Every group effect  
we could ever want  
to know... 😊

**For comparison with Fixed Quadratic, Random Linear Time Model 4b:**

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	PersonID	935.14	189.68	4.93	<.0001
UN(2,1)	PersonID	10.0839	26.9714	0.37	0.7085
UN(2,2)	PersonID	33.9319	7.6434	4.44	<.0001
UN(1,1)	ItemID	49.0094	11.8590	4.13	<.0001
Residual		16994	97.6812	173.98	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
764,041	8	764057	764057	764041	764041	764049

**Model 5b: Viewing Condition\*Quadratic Time Fixed Effects with Random Linear Time...**

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	PersonID	633.08	130.85	4.84	<.0001
UN(2,1)	PersonID	23.9355	22.1494	1.08	0.2799
UN(2,2)	PersonID	32.9677	7.4511	4.42	<.0001
UN(1,1)	ItemID	48.7616	11.7744	4.14	<.0001
Residual		16991	97.6628	173.98	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
764,008	17	764042	764042	764008	764008	764025

Better model?

**Calculate Pseudo-R<sup>2</sup> for subject random intercept variance** →  $(935.14 - 633.08) / 935.14 = 32\%$  ☺

**Calculate Pseudo-R<sup>2</sup> for subject random linear time variance** →  $(33.93 - 32.97) / 33.93 = 2.8\%$

**Calculate Pseudo-R<sup>2</sup> for item random intercept variance** →  $(49.01 - 48.76) / 49.01 = 0.5\%$

**Calculate Pseudo-R<sup>2</sup> for residual variance** →  $(16,994 - 16,991) / 16,991 = 0\%$

**R = .158 of predicted with DV,  
so total R<sup>2</sup> = 2.5%** ☹

Solution for Fixed Effects							
Effect	Condition	Estimate	Standard Error	DF	t Value	Pr >  t	simple effects
Intercept		222.45	6.9855	54.5	31.84	<.0001	
FixTime1s		9.6112	2.3425	212	4.10	<.0001	
FixTime1s*FixTime1s		-0.08287	0.3937	6E4	-0.21	0.8333	
Cond	Freeview	37.8970	10.2330	53.5	3.70	0.0005	
Cond	Memory	28.8654	10.0177	53.4	2.88	0.0057	
Cond	Pleasantness	0.5851	9.8248	53.3	0.06	0.9527	
Cond	Search	0	.	.	.	.	
FixTime1s*Cond	Freeview	5.8220	3.5001	224	1.66	0.0976	
FixTime1s*Cond	Memory	5.1632	3.4297	226	1.51	0.1336	
FixTime1s*Cond	Pleasantness	8.2860	3.3511	221	2.47	0.0142	
FixTime1s*Cond	Search	0	.	.	.	.	
FixTime1s*FixTime1s*Cond	Freeview	-1.6677	0.5910	61E3	-2.82	0.0048	
FixTime1s*FixTime1s*Cond	Memory	-1.3643	0.5790	61E3	-2.36	0.0185	
FixTime1s*FixTime1s*Cond	Pleasantness	-1.6688	0.5641	61E3	-2.96	0.0031	
FixTime1s*FixTime1s*Cond	Search	0	.	.	.	.	

Type 3 Tests of Fixed Effects **omnibus group effects**

Effect	Num DF	Den DF	F Value	Pr > F
FixTime1s	1	233	136.11	<.0001
FixTime1s*FixTime1s	1	59E3	35.20	<.0001
Cond	3	53.6	7.29	0.0003
FixTime1s*Cond	3	230	2.16	0.0931
FixTime1s*FixTime1s*Cond	3	61E3	3.96	0.0078

## Estimates FUN with follow-ups!

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at 1sec: Free-View	260.35	7.5745	55.3	34.37	<.0001
Intercept at 1sec: Memory	251.32	7.2810	55.4	34.52	<.0001
Intercept at 1sec: Pleasantness	223.04	7.0132	55.3	31.80	<.0001
Intercept at 1sec: Search	222.45	6.9855	54.5	31.84	<.0001
Intercept at 1sec: Free-View vs Memory	-9.0315	10.4370	53.9	-0.87	0.3907
<b>Intercept at 1sec: Free-View vs Pleasantness</b>	<b>-37.3119</b>	<b>10.2519</b>	<b>53.8</b>	<b>-3.64</b>	<b>0.0006</b>
<b>Intercept at 1sec: Free-View vs Search</b>	<b>-37.8970</b>	<b>10.2330</b>	<b>53.5</b>	<b>-3.70</b>	<b>0.0005</b>
<b>Intercept at 1sec: Memory vs Pleasantness</b>	<b>-28.2804</b>	<b>10.0370</b>	<b>53.8</b>	<b>-2.82</b>	<b>0.0068</b>
<b>Intercept at 1sec: Memory vs Search</b>	<b>-28.8654</b>	<b>10.0177</b>	<b>53.4</b>	<b>-2.88</b>	<b>0.0057</b>
Intercept at 1sec: Pleasantness vs Search	-0.5851	9.8248	53.3	-0.06	0.9527
<b>Linear at 1sec: Free-View</b>	<b>15.4332</b>	<b>2.6056</b>	<b>237</b>	<b>5.92</b>	<b>&lt;.0001</b>
<b>Linear at 1sec: Memory</b>	<b>14.7744</b>	<b>2.5094</b>	<b>240</b>	<b>5.89</b>	<b>&lt;.0001</b>
<b>Linear at 1sec: Pleasantness</b>	<b>17.8972</b>	<b>2.4022</b>	<b>233</b>	<b>7.45</b>	<b>&lt;.0001</b>
<b>Linear at 1sec: Search</b>	<b>9.6112</b>	<b>2.3425</b>	<b>212</b>	<b>4.10</b>	<b>&lt;.0001</b>
Linear at 1sec: Free-View vs Memory	-0.6588	3.6153	238	-0.18	0.8556
Linear at 1sec: Free-View vs Pleasantness	2.4640	3.5410	234	0.70	0.4872
Linear at 1sec: Free-View vs Search	-5.8220	3.5001	224	-1.66	0.0976
Linear at 1sec: Memory vs Pleasantness	3.1228	3.4713	236	0.90	0.3693
Linear at 1sec: Memory vs Search	-5.1632	3.4297	226	-1.51	0.1336
<b>Linear at 1sec: Pleasantness vs Search</b>	<b>-8.2860</b>	<b>3.3511</b>	<b>221</b>	<b>-2.47</b>	<b>0.0142</b>
<b>Quadratic for all Time: Free-View</b>	<b>-1.7505</b>	<b>0.4448</b>	<b>6E4</b>	<b>-3.94</b>	<b>&lt;.0001</b>
<b>Quadratic for all Time: Memory</b>	<b>-1.4471</b>	<b>0.4276</b>	<b>61E3</b>	<b>-3.38</b>	<b>0.0007</b>
<b>Quadratic for all Time: Pleasantness</b>	<b>-1.7517</b>	<b>0.4086</b>	<b>6E4</b>	<b>-4.29</b>	<b>&lt;.0001</b>
Quadratic for all Time: Search	-0.08287	0.3937	6E4	-0.21	0.8333
Quadratic for all Time: Free-View vs Memory	0.3034	0.6154	61E3	0.49	0.6220
Quadratic for all Time: Free-View vs Pleasantness	-0.00117	0.6017	6E4	-0.00	0.9985
<b>Quadratic for all Time: Free-View vs Search</b>	<b>1.6677</b>	<b>0.5910</b>	<b>61E3</b>	<b>2.82</b>	<b>0.0048</b>
Quadratic for all Time: Memory vs Pleasantness	-0.3046	0.5897	61E3	-0.52	0.6055
<b>Quadratic for all Time: Memory vs Search</b>	<b>1.3643</b>	<b>0.5790</b>	<b>61E3</b>	<b>2.36</b>	<b>0.0185</b>
<b>Quadratic for all Time: Pleasantness vs Search</b>	<b>1.6688</b>	<b>0.5641</b>	<b>61E3</b>	<b>2.96</b>	<b>0.0031</b>

